NAG Fortran Library Routine Document

F08PBF (DGEESX)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08PBF (DGEESX) computes the eigenvalues, the real Schur form T, and, optionally, the matrix of Schur vectors Z for an n by n real nonsymmetric matrix A.

2 Specification

```
      SUBROUTINE F08PBF (JOBVS, SORT, SELECT, SENSE, N, A, LDA, SDIM, WR, WI,
     1                   VS, LDVS, RCONDE, RCONDV, WORK, LWORK, IWORK, LIWORK,
     2                   BWORK, INFO)
      INTEGER            N, LDA, SDIM, LDVS, LWORK, IWORK(*), LIWORK, INFO
      double precision   A(LDA,*), WR(*), WI(*), VS(LDVS,*), RCONDE, RCONDV,
     1                   WORK(*)
      LOGICAL            SELECT, BWORK(*)
      CHARACTER*1        JOBVS, SORT, SENSE
      EXTERNAL           SELECT
```

The routine may be called by its LAPACK name ***dgeesx***.

3 Description

The real Schur factorization of A is given by

$$A = ZTZ^T,$$

where Z is orthogonal, the matrix of Schur vectors, and T is quasi upper-triangular with 1 by 1 and 2 by 2 diagonal blocks.

A real matrix is in real Schur form if it is upper quasi-triangular with 1 by 1 and 2 by 2 blocks. 2 by 2 blocks will be standardized in the form

$$\begin{bmatrix} a & b \\ c & a \end{bmatrix}$$

where $bc < 0$. The eigenvalues of such a block are $a \pm \sqrt{bc}$.

Optionally, F08PBF (DGEESX) also orders the eigenvalues on the diagonal of the real Schur form so that selected eigenvalues are at the top left; computes a reciprocal condition number for the average of the selected eigenvalues (RCONDE); and computes a reciprocal condition number for the right invariant subspace corresponding to the selected eigenvalues (RCONDV). The leading columns of Z form an orthonormal basis for this invariant subspace.

For further explanation of the reciprocal condition numbers RCONDE and RCONDV, see Section 4.8 of Anderson *et al.* (1999) (where these quantities are called s and sep respectively).

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia URL: http://www.netlib.org/lapack/lug

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

1: JOBVS – CHARACTER*1 *Input*

On entry: if JOBVS = 'N', Schur vectors are not computed.

If JOBVS = 'V', Schur vectors are computed.

2: SORT – CHARACTER*1 *Input*

On entry: specifies whether or not to order the eigenvalues on the diagonal of the Schur form:

if SORT = 'N', eigenvalues are not ordered;
if SORT = 'S', eigenvalues are ordered (see SELECT).

3: SELECT – LOGICAL FUNCTION, supplied by the user. *External Procedure*

If SORT = 'S', SELECT is used to select eigenvalues to sort to the top left of the Schur form.

If SORT = 'N', SELECT is not referenced and F08PBF (DGEESX) may be called with the dummy function F08PAZ.

Its specification is:

```
LOGICAL FUNCTION SELECT (WR, WI)
double precision          WR, WI
```

1: WR – ***double precision*** *Input*
2: WI – ***double precision*** *Input*

On entry: an eigenvalue $\mathrm{WR}(j) + \sqrt{-1} \times \mathrm{WI}(j)$ is selected if $\mathrm{SELECT}(\mathrm{WR}(j), \mathrm{WI}(j))$ is .TRUE.. If either one of a complex conjugate pair of eigenvalues is selected, then both are. Note that a selected complex eigenvalue may no longer satisfy $\mathrm{SELECT}(\mathrm{WR}(j), \mathrm{WI}(j)) =$ .TRUE. after ordering, since ordering may change the value of complex eigenvalues (especially if the eigenvalue is ill-conditioned); in this case INFO may be set to $\mathrm{N} + 2$ (see INFO below).

SELECT must be declared as EXTERNAL in the (sub)program from which F08PBF (DGEESX) is called. Parameters denoted as *Input* must **not** be changed by this procedure.

4: SENSE – CHARACTER*1 *Input*

On entry: determines which reciprocal condition numbers are computed:

if SENSE = 'N', none are computed;
if SENSE = 'E', computed for average of selected eigenvalues only;
if SENSE = 'V', computed for selected right invariant subspace only;
if SENSE = 'B', computed for both.

If SENSE = 'E', 'V' or 'B', SORT must equal 'S'.

5: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: $\mathrm{N} \geq 0$.

6: A(LDA,$*$) – ***double precision*** array *Input/Output*

Note: the second dimension of the array A must be at least $\max(1, \mathrm{N})$.

On entry: the n by n matrix A.

On exit: is overwritten by its Schur form T.

7: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F08PBF (DGEESX) is called.

Constraint: LDA $\geq \max(1, \mathrm{N})$.

8: SDIM – INTEGER *Output*

On exit: if SORT = 'N', SDIM = 0.

If SORT = 'S', SDIM = number of eigenvalues (after sorting) for which SELECT is true. (Complex conjugate pairs for which SELECT is true for either eigenvalue count as 2.)

9: WR($*$) – ***double precision*** array *Output*

Note: the dimension of the array WR must be at least $\max(1, \mathrm{N})$.

On exit: see the description of WI below.

10: WI($*$) – ***double precision*** array *Output*

Note: the dimension of the array WI must be at least $\max(1, \mathrm{N})$.

On exit: WR and WI contain the real and imaginary parts, respectively, of the computed eigenvalues in the same order that they appear on the diagonal of the output Schur form T. Complex conjugate pairs of eigenvalues will appear consecutively with the eigenvalue having the positive imaginary part first.

11: VS(LDVS,$*$) – ***double precision*** array *Output*

Note: the second dimension of the array VS must be at least $\max(1, \mathrm{N})$ if JOBVS = 'V' and at least 1 otherwise.

On exit: if JOBVS = 'V', VS contains the orthogonal matrix Z of Schur vectors.

If JOBVS = 'N', VS is not referenced.

12: LDVS – INTEGER *Input*

On entry: the first dimension of the array VS as declared in the (sub)program from which F08PBF (DGEESX) is called.

Constraints:

if JOBVS = 'V', LDVS $\geq \max(1, \mathrm{N})$;
LDVS ≥ 1 otherwise.

13: RCONDE – ***double precision*** *Output*

On exit: if SENSE = 'E' or 'B', contains the reciprocal condition number for the average of the selected eigenvalues.

If SENSE = 'N' or 'V', RCONDE is not referenced.

14: RCONDV – ***double precision*** *Output*

On exit: if SENSE = 'V' or 'B', RCONDV contains the reciprocal condition number for the selected right invariant subspace.

If SENSE = 'N' or 'E', RCONDV is not referenced .

15: WORK($*$) – ***double precision*** array *Workspace*

Note: the dimension of the array WORK must be at least $\max(1, \mathrm{LWORK})$.

On exit: if INFO = 0, WORK(1) returns the optimal LWORK.

16: LWORK – INTEGER *Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which F08PBF (DGEESX) is called.

For good performance, LWORK must generally be larger than the minimum; increase the workspace by, say, $nb \times \mathrm{N}$, where nb is the optimal block size for F08NEF (DGEHRD).

If $\mathrm{LWORK} = -1$, a workspace query is assumed; the routine only calculates upper bounds on the optimal sizes of the arrays WORK and IWORK, returns these values as the first entries of the WORK and IWORK arrays, and no error messages related to LWORK or LIWORK is issued.

If SENSE = 'E', 'V' or 'B', $\mathrm{LWORK} \geq \mathrm{N} + 2 \times \mathrm{SDIM} \times (\mathrm{N} - \mathrm{SDIM})$, where SDIM is the number of selected eigenvalues computed by this routine. Note that $\mathrm{N} + 2 \times \mathrm{SDIM} \times (\mathrm{N} - \mathrm{SDIM}) \leq \mathrm{N} + \mathrm{N} \times \mathrm{N}/2$.

Note also that an error is only returned if $\mathrm{LWORK} < \max(1, 3 \times \mathrm{N})$, but if SENSE = 'E', 'V' or 'B' this may not be large enough.

Constraint: $\mathrm{LWORK} \geq \max(1, 3 \times \mathrm{N})$.

17: IWORK($*$) – INTEGER array *Workspace*

Note: the dimension of the array IWORK must be at least $\max(1, \mathrm{LIWORK})$.

On exit: if $\mathrm{INFO} = 0$, $\mathrm{IWORK}(1)$ returns the optimal LIWORK.

18: LIWORK – INTEGER *Input*

On entry: the dimension of the array IWORK as declared in the (sub)program from which F08PBF (DGEESX) is called.

If $\mathrm{LIWORK} = -1$, a workspace query is assumed; the routine only calculates upper bounds on the optimal sizes of the arrays WORK and IWORK, returns these values as the first entries of the WORK and IWORK arrays, and no error messages related to LWORK or LIWORK is issued.

Constraints:

if SENSE = 'V' or 'B', $\mathrm{LIWORK} \geq \mathrm{SDIM} \times (\mathrm{N} - \mathrm{SDIM})$;
$\mathrm{LIWORK} \geq 1$ otherwise.

Note: $\mathrm{SDIM} \times (\mathrm{N} - \mathrm{SDIM}) \leq \mathrm{N} \times \mathrm{N}/4$. Note also that an error is only returned if $\mathrm{LIWORK} < 1$, but if SENSE = 'V' or 'B' this may not be large enough.

19: BWORK($*$) – LOGICAL array *Workspace*

Note: the dimension of the array BWORK must be at least $\max(1, \mathrm{N})$ if SORT $\neq$ 'N' and at least 1 otherwise.

If SORT = 'N', BWORK is not referenced.

20: INFO – INTEGER *Output*

On exit: $\mathrm{INFO} = 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

$\mathrm{INFO} < 0$

If $\mathrm{INFO} = -i$, the ith argument had an illegal value.

$\mathrm{INFO} > 0$

If $\mathrm{INFO} = i$ and $i \leq \mathrm{N}$, the QR algorithm failed to compute all the eigenvalues

INFO $= N + 1$

The eigenvalues could not be reordered because some eigenvalues were too close to separate (the problem is very ill-conditioned).

INFO $= N + 2$

After reordering, roundoff changed values of some complex eigenvalues so that leading eigenvalues in the Schur form no longer satisfy SELECT = .TRUE.. This could also be caused by underflow due to scaling.

7 Accuracy

The computed Schur factorization satisfies

$$A + E = ZTZ^T,$$

where

$$\|E\|_2 = O(\epsilon)\|A\|_2,$$

and ϵ is the ***machine precision***. See Section 4.8 of Anderson *et al.* (1999) for further details.

8 Further Comments

The total number of floating-point operations is proportional to n^3.

The complex analogue of this routine is F08PPF (ZGEESX).

9 Example

To find the Schur factorization of the matrix

$$A = \begin{pmatrix} 0.35 & 0.45 & -0.14 & -0.17 \\ 0.09 & 0.07 & -0.54 & 0.35 \\ -0.44 & -0.33 & -0.03 & 0.17 \\ 0.25 & -0.32 & -0.13 & 0.11 \end{pmatrix},$$

such that the real eigenvalues of A are the top left diagonal elements of the Schur form, T. Estimates of the condition numbers for the selected eigenvalue cluster and corresponding invariant subspace are also returned.

Note that the block size (NB) of 64 assumed in this example is not realistic for such a small problem, but should be suitable for large problems.

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F08PBF Example Program Text
*     Mark 21.  NAG Copyright 2004.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NB, NMAX
      PARAMETER        (NB=64,NMAX=10)
      INTEGER          LDA, LDVS, LIWORK, LWORK
      PARAMETER        (LDA=NMAX,LDVS=NMAX,LIWORK=(NMAX*NMAX)/4,
     +                 LWORK=NMAX*(NB+2+NMAX/2))
*     .. Local Scalars ..
      DOUBLE PRECISION ANORM, EPS, RCONDE, RCONDV, TOL
      INTEGER          I, IFAIL, INFO, J, LWKOPT, N, SDIM
*     .. Local Arrays ..
      DOUBLE PRECISION A(LDA,NMAX), VS(LDVS,NMAX), WI(NMAX),
```

```
      +                   WORK(LWORK), WR(NMAX)
       INTEGER            IWORK(LIWORK)
       LOGICAL            BWORK(NMAX)
*      .. External Functions ..
       DOUBLE PRECISION F06RAF, X02AJF
       LOGICAL            SELECT
       EXTERNAL           F06RAF, X02AJF, SELECT
*      .. External Subroutines ..
       EXTERNAL           DGEESX, X04CAF
*      .. Executable Statements ..
       WRITE (NOUT,*) 'F08PBF Example Program Results'
       WRITE (NOUT,*)
*      Skip heading in data file
       READ (NIN,*)
       READ (NIN,*) N
       IF (N.LE.NMAX) THEN
*
*         Read in the matrix A
*
          READ (NIN,*) ((A(I,J),J=1,N),I=1,N)
*
*         Find the Frobenius norms of A
*
          ANORM = F06RAF('Frobenius',N,N,A,LDA,WORK)
*
*         Find the Schur factorization of A
*
          CALL DGEESX('Vectors (Schur)','Sort',SELECT,
      +               'Both reciprocal condition numbers',N,A,LDA,SDIM,
      +               WR,WI,VS,LDVS,RCONDE,RCONDV,WORK,LWORK,IWORK,
      +               LIWORK,BWORK,INFO)
          LWKOPT = WORK(1)
*
          IF (INFO.EQ.0 .OR. INFO.EQ.(N+2)) THEN
*
*            Print solution
*
             WRITE (NOUT,99999)
      +        'Number of eigenvalues for which SELECT is true = ', SDIM,
      +        '(dimension of invariant subspace)'
             WRITE (NOUT,*)
             IF (INFO.EQ.(N+2)) THEN
                WRITE (NOUT,99998) '***Note that rounding errors mean ',
      +           'that leading eigenvalues in the Schur form',
      +           'no longer satisfy SELECT = .TRUE.'
                WRITE (NOUT,*)
             END IF
*
*            Print out factors of the Schur factorization
*
             IFAIL = 0
             CALL X04CAF('General',' ',N,N,A,LDA,'Schur matrix T',IFAIL)
*
             WRITE (NOUT,*)
             CALL X04CAF('General',' ',N,N,VS,LDVS,
      +                   'Matrix of Schur vectors Z',IFAIL)
*
*            Print out the reciprocal condition numbers
*
             WRITE (NOUT,*)
             WRITE (NOUT,99997)
      +        'Reciprocal of projection norm onto the invariant',
      +        'subspace for the selected eigenvalues', 'RCONDE = ',
      +        RCONDE
             WRITE (NOUT,*)
             WRITE (NOUT,99996)
      +        'Reciprocal condition number for the invariant subspace',
      +        'RCONDV = ', RCONDV
*
*            Compute the machine precision
*
```

```
            EPS = X02AJF()
            TOL = EPS*ANORM
*
*           Print out the approximate asymptotic error bound on the
*           average absolute error of the selected eigenvalues given by
*
*              eps*norm(A)/RCONDE
*
            WRITE (NOUT,*)
            WRITE (NOUT,99995)
     +        'Approximate asymptotic error bound for selected ',
     +        'eigenvalues   = ', TOL/RCONDE
*
*           Print out an approximate asymptotic bound on the maximum
*           angular error in the computed invariant subspace given by
*
*              eps*norm(A)/RCONDV
*
            WRITE (NOUT,99995)
     +        'Approximate asymptotic error bound for the invariant ',
     +        'subspace = ', TOL/RCONDV
         ELSE
            WRITE (NOUT,99994) 'Failure in DGEESX. INFO =', INFO
         END IF
*
*        Print workspace information
*
         IF (LWORK.LT.LWKOPT) THEN
            WRITE (NOUT,*)
            WRITE (NOUT,99993) 'Optimum workspace required = ', LWKOPT,
     +        'Workspace provided         = ', LWORK
         END IF
      ELSE
         WRITE (NOUT,*)
         WRITE (NOUT,*) 'NMAX too small'
      END IF
      STOP
*
99999 FORMAT (1X,A,I4,/1X,A)
99998 FORMAT (1X,2A,/1X,A)
99997 FORMAT (1X,A,/1X,A,/1X,A,1P,E8.1)
99996 FORMAT (1X,A,/1X,A,1P,E8.1)
99995 FORMAT (1X,2A,1P,E8.1)
99994 FORMAT (1X,A,I4)
99993 FORMAT (1X,A,I5,/1X,A,I5)
      END

      LOGICAL FUNCTION SELECT(WR,WI)
*     .. Scalar Arguments ..
*
*     Logical function SELECT for use with DGEESX (F08PBF)
*
*     Returns the value .TRUE. if the imaginary part of the eigenvalue
*     (WR + WI*i) is zero, i.e. the eigenvalue is real
*
      DOUBLE PRECISION        WI, WR
*     .. Local Scalars ..
      LOGICAL                 D
*     .. Executable Statements ..
      IF (WI.EQ.0.0D0) THEN
         D = .TRUE.
      ELSE
         D = .FALSE.
      END IF
*
      SELECT = D
*
      RETURN
      END
```

9.2 Program Data

```
F08PBF Example Program Data

  4                         :Value of N

  0.35   0.45  -0.14  -0.17
  0.09   0.07  -0.54   0.35
 -0.44  -0.33  -0.03   0.17
  0.25  -0.32  -0.13   0.11 :End of matrix A
```

9.3 Program Results

```
 F08PBF Example Program Results

 Number of eigenvalues for which SELECT is true =    2
 (dimension of invariant subspace)

 Schur matrix T
          1       2       3       4
 1   0.7995 -0.0059  0.0751 -0.0927
 2   0.0000 -0.1007 -0.3937  0.3569
 3   0.0000  0.0000 -0.0994  0.5128
 4   0.0000  0.0000 -0.3132 -0.0994

 Matrix of Schur vectors Z
          1       2       3       4
 1   0.6551  0.1210 -0.5032 -0.5504
 2   0.5236  0.3286  0.7857 -0.0229
 3  -0.5362  0.5974  0.0904 -0.5894
 4   0.0956  0.7215 -0.3482  0.5908

 Reciprocal of projection norm onto the invariant
 subspace for the selected eigenvalues
 RCONDE =  5.7E-01

 Reciprocal condition number for the invariant subspace
 RCONDV =  3.1E-01

 Approximate asymptotic error bound for selected eigenvalues   =  2.2E-16
 Approximate asymptotic error bound for the invariant subspace =  4.1E-16
```